NAG Fortran Library Routine Document

F08CJF (DORGRQ)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08CJF (DORGRQ) generates all or part of the real n by n orthogonal matrix Q from an RQ factorization computed by F08CHF (DGERQF).

2 Specification

```
SUBROUTINE F08CJF (M, N, K, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER            M, N, K, LDA, LWORK, INFO
double precision   A(LDA,*), TAU(*), WORK(*)
```

The routine may be called by its LAPACK name ***dorgrq***.

3 Description

F08CJF (DORGRQ) is intended to be used following a call to F08CHF (DGERQF), which performs an RQ factorization of a real matrix A and represents the orthogonal matrix Q as a product of k elementary reflectors of order n.

This routine may be used to generate Q explicitly as a square matrix, or to form only its trailing rows.

Usually Q is determined from the RQ factorization of a p by n matrix A with $p \leq n$. The whole of Q may be computed by:

```
CALL DORGRQ (N,N,P,A,LDA,TAU,WORK,LWORK,INFO)
```

(note that the array A must have at least n rows), or its trailing p rows as:

```
CALL DORGRQ (P,N,P,A,LDA,TAU,WORK,LWORK,INFO)
```

The rows of Q returned by the last call form an orthonormal basis for the space spanned by the rows of A; thus F08CHF (DGERQF) followed by F08CJF (DORGRQ) can be used to orthogonalise the rows of A.

The information returned by F08CHF (DGERQF) also yields the RQ factorization of the trailing k rows of A, where $k < p$. The orthogonal matrix arising from this factorization can be computed by:

```
CALL DORGRQ (N,N,K,A,LDA,TAU,WORK,LWORK,INFO)
```

or its leading k columns by:

```
CALL DORGRQ (K,N,K,A,LDA,TAU,WORK,LWORK,INFO)
```

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix Q.

Constraint: $M \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix Q.

Constraint: $N \geq M$.

3: K – INTEGER *Input*

On entry: k, the number of elementary reflectors whose product defines the matrix Q.

Constraint: $M \geq K \geq 0$.

4: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the $(m - k + i)$th row must contain the vector which defines the elementary reflector H_i, for $i = 1, 2, \ldots, k$, as returned by F08CHF (DGERQF) in the last k rows of its array argument A.

On exit: the m by n matrix Q.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08CJF (DORGRQ) is called.

Constraint: $LDA \geq \max(1, M)$.

6: TAU($*$) – ***double precision*** array *Input*

Note: the dimension of the array TAU must be at least $\max(1, K)$.

On entry: TAU(i) must contain the scalar factor of the elementary reflector H_i, as returned by F08CHF (DGERQF).

7: WORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, LWORK)$.

On exit: if INFO $= 0$, WORK(1) contains the minimum value of LWORK required for optimal performance.

8: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08CJF (DORGRQ) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, $LWORK \geq N \times nb$, where nb is the optimal ***block size***.

Constraint: $LWORK \geq \max(1, M)$ or LWORK $= -1$.

9: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed matrix Q differs from an exactly orthogonal matrix by a matrix E such that

$$\|E\|_2 = O\,\epsilon$$

and ϵ is the ***machine precision***.

8 Further Comments

The total number of floating point operations is approximately $4mnk-2(m+n)k^2+\frac{4}{3}k^3$; when $m = k$ this becomes $\frac{2}{3}m^2(3n-m)$.

The complex analogue of this routine is F08CWF (ZUNGRQ).

9 Example

This example generates the first four rows of the matrix Q of the RQ factorization of A as returned by F08CHF (DGERQF), where

$$A = \begin{pmatrix} -0.57 & -1.93 & 2.30 & -1.93 & 0.15 & -0.02 \\ -1.28 & 1.08 & 0.24 & 0.64 & 0.30 & 1.03 \\ -0.39 & -0.31 & 0.40 & -0.66 & 0.15 & -1.43 \\ 0.25 & -2.14 & -0.35 & 0.08 & -2.13 & 0.50 \end{pmatrix}.$$

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

```
*     F08CJF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NMAX
      PARAMETER        (MMAX=8,NB=64,NMAX=8)
      INTEGER          LDA, LWORK
      PARAMETER        (LDA=MMAX,LWORK=NB*MMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, M, N
      CHARACTER*26     TITLE
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), TAU(NMAX), WORK(LWORK)
*     .. External Subroutines ..
      EXTERNAL         DGERQF, DORGRQ, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08CJF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.LE.N) THEN
*
*        Read A from data file
*
```

```
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
*
*        Compute the RQ factorization of A
*
         CALL DGERQF(M,N,A,LDA,TAU,WORK,LWORK,INFO)
*
*        Form the leading M rows of Q explicitly
*
         CALL DORGRQ(M,N,M,A,LDA,TAU,WORK,LWORK,INFO)
*
*        Form the heading for X04CAF
*
         WRITE (TITLE,99999) M
*
*        Print the leading M rows of Q
*
         IFAIL = 0
         CALL X04CAF('General',' ',M,N,A,LDA,TITLE,IFAIL)
*
      ELSE
         WRITE (NOUT,*) 'MMAX and/or NMAX is too small, and/or M.GT.N'
      END IF
      STOP
*
99999 FORMAT ('The leading ',I4,' rows of Q')
      END
```

9.2 Program Data

```
F08CJF Example Program Data

  4      6                                :Values of M and N

 -0.57  -1.93   2.30  -1.93   0.15  -0.02
 -1.28   1.08   0.24   0.64   0.30   1.03
 -0.39  -0.31   0.40  -0.66   0.15  -1.43
  0.25  -2.14  -0.35   0.08  -2.13   0.50 :End of matrix A
```

9.3 Program Results

```
 F08CJF Example Program Results

 The leading    4 rows of Q
          1       2       3       4       5       6
 1  -0.0833  0.2972 -0.6404  0.4461 -0.2938 -0.4575
 2   0.9100 -0.1080 -0.2351 -0.1620  0.2022 -0.1946
 3  -0.2202 -0.2706  0.2220 -0.3866  0.0015 -0.8243
 4  -0.0809  0.6922  0.1132 -0.0259  0.6890 -0.1617
```